	Ein klassischer VFP-Bericht automatisiert die Ausgabe von WinWord

Lisa Slater Nicholls

In „WinWord für VFP-Berichte automatisieren“ habe ich Ihnen ein einfaches Beispiel vergestellt, wie Sie mit OLE Automation einen Bericht in WinWord erstellen können. Dieses Beispiel generiert Resultate, die auf einem einzigen VFP-Datensatz basieren. Das Schreiben von WordBasic-Code, der Resultate mehrerer Datensätze auflistet, erfordert schon einige Genialität, vor allem, wenn die Anzahl der Datensätze im Resultat variiert. Die Behandlung von Code, der unter bestimmten Bedingungen die Datensätze in Gruppen aufteilt, die eigene Überschriften und Berechnungen erfordern, macht ebenfalls Probleme. In diesem Artikel behandle ich einige der Probleme, und im guten FoxPro-Stil zeige ich Ihnen den Umgang mit einer benutzerdefinierten Klasse, die den größten Teil der Arbeit erledigt.

Ich informiere Sie über verschiedene Tricks, die Sie im Umgang mit WordBasic kennen sollten, um Ihre Zeit nicht zu verschwenden, besonders wenn Sie bereits wissen, wie Sie mehrere Datensätze in Visual FoxPro behandeln müssen. Im Grunde benutzen Sie ein SCAN ... ENDSCAN, um durch Ihre Datensätze zu gehen, und Sie betten SCAN WHILEs ein, um die Daten zu gruppieren, auf die Sie in den inneren SCANs treffen. Oder nicht?


Die guten Seiten der FRX

Es ist schon lange her, seit Sie diese Art von Bericht geschrieben haben. VFPs Berichts-Designer erledigt das Gruppieren der Daten und die Bewegungen des Satzzeigers für Sie – und diese Arbeit ist auch seine Aufgabe, besonders wenn wir entscheiden, daß eine FRX-Datei nicht in der Lage ist, unsere Resultate angemessen zu formatieren. Alles was wir brauchen ist eine Möglichkeit, die Ausgabe durch WinWord auszulösen, wie der Berichtsgenerator durch die Datensätze geht und sie gruppiert. 

Die Ereignisse von VFPs Berichtsgenerator geben uns diese Möglichkeit. In FoxPro 2.x haben wir dafür vielleicht UDFs benutzt, aber die Ereignisse in VFP-Berichten sind einfacher zu kontrollieren. Sie müssen exakt auf jedes Ereignis reagieren und dabei auch auf andere Aktionen des Berichtsgenerators achten. Für WinWord-Text, der Daten satzweise anzeigt, lassen Sie WordBasic-Code durch die Ereignisse im Detailbereich ausführen. Für zusammenfassende Informationen, die einmal für jede Gruppe ausgeführt werden sollen, schreiben Sie die WordBasic-Anweisungen in den Gruppenkopf oder –fuß. 

Das Ergebnis ist ein FRX, der keine eigene Ausgabe besitzt, wie Sie in Abbildung 1 sehen können. Die Ereignisse des FRX rufen verschiedene benutzerdefinierte Prozeduren auf, um eine Ausgabe in der von Ihnen gewünschten Form zu erzeugen.

Die Technik generisch machen

Über diese Technik können Sie glücklich sein, wenn Sie bloß einen Bericht zu erzeugen haben. Ich habe lediglich einige Gruppierungen in meinem leeren Bericht, FRXCLASS.FRX, benutzt. Dafür habe ich die höchste Anzahl Gruppierungen, die ich vermutlich jemals in einem Bericht benutzen werde, in das FRX gesetzt; wenn Sie es wünschen, können Sie auch mehr benutzen. Für jeden Bericht, der weniger Gruppierungen erfordert, gibt der Gruppenausdruck lediglich .T. zurück – und solche Gruppierungen werden nicht ausgeführt. Dementsprechend behalten Ereignisse, die ich nicht immer benötige (z.B. das OnEntry und OnExit eines Titels in einem Bericht, der keinen Titel benötigt), ihre Prozeduren für alle Berichte. Wird in einem bestimmten Bericht kein Titel benötigt, RETURNen diese Prozeduren lediglich ein .T. und keinen Inhalt.

Wie Sie wahrscheinlich bemerkt haben werden, habe ich die Ereignisprozeduren Page und Column in FRXCLASS.FRX ausgelassen. Warum habe ich nicht wie bei anderen Ereignissen leere Prozeduren hinzugefügt, wenn diese Ereignisse doch immer benötigt werden?

Anders als andere Gruppen des Berichtsgenerators sind die Adressierungen von Page und Column Teile des Physikalischen Layout. Ich will den Berichtsgenerator nicht entscheiden lassen, wann eine Seite beginnt oder endet oder wann eine Spalte umbrochen werden muß. Die unintelligente Behandlung der physikalischen Seitengröße ist eines der größten Mankos des Berichtsgenerators von Visual FoxPro. Ich möchte die Seitenbehandlung auf den Teil meines Programms einschränken, in dem ich mit der Textverarbeitung, einer anderen Anwendung oder einem Gerät kommuniziere.

Ableitung von Klassen mit dieser 
Technik

Vermutlich haben Sie bereits ausgeknobelt, daß es nicht notwendig ist, diese UDFs für jeden WinWord-Bericht, den Sie erstellen, neu zu schreiben. Sie könnten beispielsweise ein Rahmenprogramm haben, das FRXCLASS.FRX aufruft, und das in etwa folgendermaßen aussieht:

oWord = CREATEOBJECT(Word.Basic)

REPORT FORM FrxClass

PROC DetailEntry

<Jede Menge Anweisungen>

ENDPROC

<Andere Prozeduren>

Sie könnten diesen Rahmen für jeden Bericht neu erstellen. Wenn Sie aber feststellen, daß Sie Code erneut schreiben, sollten Sie daran denken, den Code in eine Klasse zu schreiben, um ihn einfacher wiederverwenden zu können. Auch wenn dieser Code einfach zu schreiben ist, ermöglicht es doch die Erstellung von Klassen vielen Berichten, von anderen zu erben.

Zum Beispiel könnten Sie vorhaben, daß sich alle Berichte eines bestimmten Kunden bestimmte Stilelemente teilen, die Sie in einem Makro in der Eigenschaft SummaryExit plaziert haben. Wir werden das im nächsten Beispiel durchführen. Sie haben vor, das Makro ohne Cut-and-Paste mehrfach zu verwenden. Sie haben auch vor, das Schreiben leerer Prozeduren für alle Ereignisse und Gruppierungen, die Sie in dem speziellen Bericht nicht verwenden wollen, zu vermeiden. Damit vermeiden Sie auch Fehlermeldungen, daß eine Prozedur nicht gefunden wurde, wenn FRXCLASS.FRX auf die Namen dieser Prozeduren trifft.

Abbildung 1
FrxClass ist ein scheinbar leerer Bericht, der aber die Möglichkeiten für eine unbegrenzte Ausgabe enthält.
Damit haben Sie ein Problem. Die Berichte von VFP sind keine Objekte. Wie wollen Sie davon also eine Klasse ableiten und die Methoden wiederverwenden? Vielleicht sehen Sie die Antwort bereits in Abbildung 1: Jedes Ereignis ruft statt einer UDF eine getrennte Methode eines Berichtsverwaltungsobjekts auf. Der Code, der Ihren Bericht FrxClass ausführt, sieht dann ungefähr so aus:

oFRX = CREATEOBJECT(<MeinBericht>)

REPORT FORM FrxClass

DEFINE CLASS ThisReport AS <FRXMgrClass>


PROC DetailEntry


*
<ganz viel Code für die Details>


ENDPROC

ENDDEFINE

Nicht benötigte Methoden werden von der Elternklasse geerbt, so daß Sie sich keine Gedanken über die Wiederholung irgendwelcher Methoden machen müssen. 

*Auszüge aus JUGGLER1.PRG

LPARAMETERS tcAlias, tcScope, tnSession, tcTarget, tcDot 

* Fügen Sie jeden Parameter, den Sie benötigen hinzu, 

* einschließend neuer Features für die Initialisierung,

* die Sie Ihrer abgeleiteten Klasse übergeben wollen.

* die Klasse bekanntmachen:

SET PROC TO Frx2Doc.PRG ADDITIVE

IF EMPTY(tcDot) OR TYPE("tcDot") # "C"

   tcDot = "JUGGLERS.DOT"

ENDIF   

* Objektreferenz erstellen:

PUBLIC oFRX

oFRX = CREATEOBJECT("Juggler1", ;

                     tcAlias, tcScope, ;

                     tnSession, tcTarget, tcDot)

IF TYPE("oFRX") = "O" AND NOT ISNULL(oFRX)

   oFRX.Run()

ENDIF

RELEASE PROC FRX2DOC

CLEAR CLASS Juggler1

CLEAR CLASS Frx2Doc

CLEAR CLASS FrxClass

RETURN

DEFINE CLASS Juggler1 AS Frx2DOC

   cOldAlias = ALIAS()

   lSourceTableUsed = USED("Jugglers")

   PROC DEOpenTables(toDE)

      * Die Erstellung der Detail- und

      * Gruppencursor, die die Ergebnisse

      * enthalten, habe ich hier weggelassen

   ENDPROC

   PROC TitleEntry



THIS.oWordRef.InsertPara

      THIS.oWordRef.FormatStyle("Heading 1")      

      THIS.oWordRef.CenterPara

      THIS.oWordRef.Insert("Current Standings")

   ENDPROC

PROC GroupLevel1

   RETURN Detail.Name

   PROC GroupHeader1Entry

      * Ein Zeichen ausrücken 

      THIS.oWordRef.CharRight(0)

      THIS.oWordRef.InsertPara

      THIS.oWordRef.FormatStyle("Heading 2")

      THIS.oWordRef.Insert(Detail.Name)

      THIS.oWordRef.InsertPara          

      THIS.oWordRef.TableInsertTable(,3,1)

      THIS.oWordRef.TableSelectRow

      THIS.oWordRef.FormatStyle("Heading 3")      



THIS.oWordRef.TableColumnWidth('2"')      

      THIS.oWordRef.StartofRow      

      THIS.oWordRef.Insert("Trick")

      THIS.oWordRef.NextCell

      THIS.oWordRef.Insert("Date")


* Weiter mit den Einträgen für die 


* Gruppendaten

ENDPROC

PROC DetailEntry


* Aufruf der Thermometer-Methode


* wenn Sie wollen, einmal pro Gruppe

      THIS.UpdateMessage()

      * Weggelassen: Bewegen durch die

      * Detaildaten, Einfügen der Daten

ENDPROC

PROC GroupFooter1Entry

* Weggelassen: Formatierung der Fußzeile

* Das Einfügen der Informationen

* endet wie folgt

THIS.oWordRef.Insert(TRANS(Group1.SumPoints,"@( 999.99"))

THIS.oWordRef.CharRight(0)

* Ende der Tabelle

THIS.oWordRef.LineDown

ENDPROC

PROC DECloseTables(toDE)

*
Hier weggelassen: Schließen der Tabellen,


*
zurück zum Original-Alias

   ENDPROC      

PROC SummaryExit

THIS.oWordRef.FilePrintSetup("")

#DEFINE PRT_COLOR        9

      #DEFINE PRTCOLOR_COLOR   2

      IF PRTINFO(PRT_COLOR) = PRTCOLOR_COLOR

         THIS.oWordRef.ToolsMacro("UseRedForNegativeNumbers",1)

      ELSE

         THIS.oWordRef.ToolsMacro("UseBoldForNegativeNumbers",1)

      ENDIF

   ENDPROC

ENDDEFINE

Listing A zeigt Ihnen einen Auszug aus JUGGLER1.PRG. Der Code wird zur Erstellung komplexer Berichte verwendet und benötigt FRXCLASS.FRX und sein Managerobjekt Juggler1. Den fertig erstellten Bericht sehen Sie in Abbildung 2. Ich werde Ihnen im nächsten Abschnitt die Klassenhierarchie erläutern, auf der das Managerobjekt basiert. Lassen Sie uns aber erst einen Blick auf die Funktionalität von Frx2Doc, die Elternklasse von Juggler1, werfen.

Tricks in der Vorlage

Das Beispielprogramm JUGGLER1.PRG erhält zahlreiche Parameter, die alle von Frx2Doc initialisiert werden. Das einzige Argument, das direkt einen Wert bekommt, ist die Angabe, welche Formatvorlage für das Word-Dokument verwendet werden soll. Ich wollte die Bedeutung der von WinWords .DOT-Datei für den Vorgang des Berichts betonen. Dem ausgegebenen Dokument sollte eine Formatvorlage zugeordnet sein. In diesem Fall ist es JUGGLERS.DOT, der Briefkopf der Vereinigung. Sie sollten aber auch in der Lage sein, die Vorlage in den unterschiedlichen Instanzen zu überschreiben – daher übergebe ich die Datei dem PRG als Parameter.

Die Formatvorlage enthält eine ganze Anzahl Makros und Stilelemente, die das Dokument brauchen könnte. Warum ist es dann aber nicht gefährlich, die Vorlage zu überschreiben? Word besitzt ein Vererbungsschema für Formatvorlagen, das sich wie eine echte objektorientierte Klassenhierarchie verhält. Alle Formatvorlagen erben die Einstellungen der NORMAL.DOT, der globalen Vorlage. Die Vorlagen können auf anderen DOTs basieren und erben die Einstellungen ihrer Eltern.

Andere Formatvorlagen können in WinWord als AddIns geladen werden und zusätzliche Stilelemente und Makros anbieten. Diese Tatsache unterliegt einem sehr mächtigen Teil meiner VFP-WinWord-Strategie: Die Makros, die ich in JUGGLER1 benutze, kommen von FRX2DOC.DOT, die ich separat von der Juggler-Vorlagen-Hierarchie lade. FRX2DOC.DOT enthält alle Stilelemente, Makros und andere Features, auf denen der Code von Frx2Doc basiert. Dadurch steht es dem Anwender frei, die Vorlage des Briefkopfes zu gestalten und zu ändern. 

Zur Sicherheit verwende ich noch eine andere Formatvorlage, die in diesen Beispielen nicht direkt verwendet wird: FRXBASE.DOT. Sie ist komplett leer, und wird von Frx2Doc benutzt, um eine aktuelle Vorlage anzugeben, für den Fall, daß dies weder von der abgeleiteten Klasse noch vom Berichtsprogramm erledigt wird. Im Praxiseinsatz mache ich die FRXBASE.DOT den Anwendern zugänglich, damit diese alle ihre Vorlagen für Berichte mit wenigen globalen Änderungen von einer Quelle ableiten können. Ich habe schreibgeschützte Kopie in meiner APP, die bei Bedarf auf die Platte kopiert werden kann. Auch die FRX2DOC.DOT ist in der APP gespeichert und wird, bevor sie vom Frx2Doc-Prozeß geladen wird, auf die Festplatte kopiert.

Die Behandlung der Daten

Die Klasse Juggler1 benutzt eine Methode der Datenumgebung des Berichts, DEOpenTables(), um dem Bericht die erforderlichen Daten liefern zu können. Alle Ereignisse der FRX, die mit der Datenumgebung zusammenarbeiten, haben entsprechende Frx2Doc-Methoden, die eine Referenz auf die Datenumgebung als Argument erhalten. Im Code von Juggler1 mache ich mit diesem Code nichts; ich benutze lediglich die Ereignisse DEOpenTables() und DECloseTables() als einen passenden Platz, um SELECTs und anderen Code für die Behandlung der Daten für diesen Bericht unterzubringen. Ich gehe davon aus, daß diese Ereignisse zu den richtigen Zeiten eintreten.

Haben Sie bemerkt, daß FRXCLASS.FRX keine private Datensitzung benutzt? Für einen Bericht, der mehrfach verwendet werden soll, halte ich das Verbleiben in der aktuellen Datensitzung für sinnvoll. Sie könnten die FRX aber auch in einem komplexen Abfrage-Dialog einsetzen und den Dialog die Views oder Cursor erstellen lassen, aus denen Sie dann die Daten für Ihren Bericht entnehmen. Diesem Dialog können sie, falls notwendig, eine Referenz auf die entsprechende Datensitzung übergeben.

Dieser Bericht ist nach den Mitgliedern der Code Jugglers gruppiert. Vergleichen Sie dazu in Listing A die Prozedur GroupLevel1. Die Daten in JUGGLER1.PRG kommen von zwei SELECT-Anweisungen, die im Listing nicht enthalten sind. Eine Anweisung erstellt die Detaildatensätze, die andere bildet Untergruppen für jedes Mitglied. Statt eines zweiten Cursor habe ich einfach eine Variable verwendet, die ich in einem der Detail-Ereignisse aufrechne und die ich auf Null setze, wenn ich in eine neue Gruppe komme. Ich glaube, daß diese Methode ein wenig schneller ist, aber vielleicht haben Sie ja andere Prioritäten.

Abbildung 1 Wir haben eine etwas komplexere Ausgabe als WinWord-Dokument erzeugt, die mehrere Datensätze in einer Tabelle enthält.

Spezielle Features dieser Unterklasse für die Berichtsverwaltung

Geschwindigkeit ist wichtig. Sie sollten immer nach der schnellstmöglichen Methode für die Kommunikation mit Word suchen. Im Code von GroupHeader1Entry, den sie in Listing A sehen, habe ich eine Tabelle erstellt, die die Daten der aktuellen Gruppe enthält. 

Ich habe den Code für die Erstellung der Tabelle bewußt in mehrere Zeilen aufgeteilt, so daß Sie in der Lage sein sollten, zu erkennen, wie die Tabelle aussieht. Beachten sie besonders, daß ich die Spaltenbreite durch einen String festlege und dabei die Maßeinheit mit angebe (THIS.oWordRef. TableColumnWith(‘2“‘)). Sie können und sollten die Einheit immer mit angeben, da Sie nicht wissen, welches Maßsystem in WinWord voreingestellt ist. 

Auf jeden Fall vermindern OLE-Anweisungen immer die Geschwindigkeit. Wenn ich Code schreibe, der anschließend in der Praxis eingesetzt werden soll, könnte ich eine spezielle Formatvorlage erstellen, die ein Beispiel dieser Tabelle als Autotext enthält. Dann brauche ich die sie nur noch an die richtige Position einzufügen, wenn ich mit einer Gruppe beginne.

Auch mit solchen Verbesserungen bleibt die Kommunikation zwischen Komponenten recht behäbig. Sie könnten nun dem Anwender mitteilen, daß nichts schiefgegangen ist und die Verarbeitung läuft. Ich habe der Klassenhierarchie meiner Klasse FRXClass die Methode UpdateMessage() hinzugefügt, die dem Anwender in ähnlicher Form wie die Fortschrittsanzeige von WinWord mitteilt, daß seine Aktion bearbeitet wird. Sie sehen sie in Abbildung 3. In der Klasse Juggler1 habe ich UpdateMessage() nur im Detailbereich aufgerufen, da die Tabelle nur wenige Datensätze enthält; aber in größeren Tabellen kann es sinnvoll sein, sie für jede Gruppe aufzurufen.

Abbildung 3
Die Methode UpdateMessage() teilt dem Anwender mit, daß das Dokument erstellt wird.
Da Juggler1 einen Standardbriefkopf der Organisation verwendet, habe ich das Ereignis TitleEntry() benutzt, um diesem Bericht einige Formatanweisungen für diesen Bericht mitzugeben. Dieser Abschnitt kann nahezu endlos erweitert werden oder Sie speichern berichtsspezifische Informationen in einer berichtsspezifischen Formatvorlage.

Sie werden bald feststellen, daß Sie die WordBasic-Anweisungen sehr schnell schreiben können, daher ist es nicht notwendig, sie noch länger zu behandeln. Der Code, den ich in das Ereignis SummaryExit() geschrieben habe, erfordert aber doch noch einige Aufmerksamkeit. In diesem Ereignis kann ich alle Arten von globalen Formatierungen unterbringen, die ich nicht in jedem Detailbereich erneut plazieren will. Zum Beispiel formatiert der Code von SummaryExit() negative Zahlen entweder rot oder fett (je nach den Fähigkeiten des Druckers). Dafür werden Makros verwendet, die ich in FRX2DOC.DOT gespeichert habe.

Als Erstes stellt SummaryExit() Word auf den Standarddrucker ein. Sie können dem Anwender auch mit Visual FoxPros Funktion APRINTERS() eine Liste aller verfügbaren Drucker anzeigen lassen und ihn wählen lassen. Als nächstes stellen Sie mit VFPs Funktion PRTINFO() fest, ob der Drucker Farbe drucken kann, soweit VFP und Windows das feststellen können. Haben Sie APRINTERS() eingesetzt und WinWord benutzt nicht den Standarddrucker, müssen Sie PRTINFO() den optionalen zweiten Parameter mit der Druckerinformation übergeben. Anschließend können Sie entscheiden, welches Makro eingesetzt werden soll und übergeben dessen Namen WordBasics Funktion ToolsMacro(). Ich habe eines der zwei Makros, die sich im Übrigen nicht stark unterscheiden, in Listing B dargestellt.

Sub MAIN

StartOfDocument

EditFindClearFormatting

EditFind .Find = "( ", .Direction = 0, .Wrap = 0

While EditFindFount()


ExtendSelection(")")


FormatFont .Color = 6


Cancel


CharRight


EditFind .Find = "( ", .Direction = 0, .Wrap = 0

Wend

End Sub

Jenseits von Word

Sie haben nun gesehen, wie die Klasse Juggler1 auf ihrer Elternklasse, Frx2Doc, aufbaut. Frx2Doc ist eine abstrakte Klasse, die für FRXCLASS.FRX die Dokumentenstellung verwaltet. Vielleicht glauben Sie nun, das einzige komplexe wäre der Code für die generelle Behandlung von WinWord, den ich in Frx2Doc geschrieben habe. Wenn Sie eine Klassenhierarchie entwerfen und erstellen, werden Sie häufig feststellen, daß Sie das eine oder andere erneut entwerfen, bis Sie die meisten Funktionalitäten der Basisklasse in höheren Stufen der Klassenhierarchie separiert haben, die dann die unterschiedlichen Aufgaben in zusätzlichen abgeleiteten Klassen ausführen. Auch Frx2Doc war da keine Ausnahme. In der jetzigen Fassung basiert Frx2Doc auf FrxClass, einer Custom-Klasse, die FRXCLASS.FRX steuert, ohne selber eine eigene Ausgabe zu besitzen.

DEFINE CLASS Frx2Doc AS FRXClass

* überschreibt die Vorgabewerte von FRXClass:

cTarget = THIS.cFRXFileDir + "\FRX2Doc.DOC"

cTargetExt = "DOC"

Name = "FRX2DOC"

cInitMessage = "Contacting Word and initializing..."

* FRX2Doc-spezifische Eigenschaften:

cDotExt = "DOT"

oWordRef = .NULL.

cAddInFRXDOT = THIS.cFRXFileDir + "\FRX2Doc.DOT"

cDot = THIS.cFRXFileDir + "\FRXBase.DOT"

FUNC Init(tcAlias, tcScope, tnSession, tcTarget, tcDot)


IF NOT FRXClass::Init(tcAlias, tcScope, tnSession, tcTarget)

    
RETURN .F.


ENDIF


WAIT WINDOW NOWAIT THIS.cInitMessage


THIS.cDot = THIS.GetFileName(THIS.cDot,tcDot,THIS.cDotExt)


THIS.oWordRef = CREATEOBJECT("Word.Basic")


* Testen der Word-Referenz, 

    * Ausführen und Testen anderer erforderlicher 

    * Setup-Bedingungen:

    lcConditionsFailMessage = ;

    
THIS.Setup2(tcAlias,tcScope,tnSession, ;




THIS.cTarget, THIS.cDot)


IF NOT EMPTY(lcConditionsFailMessage)

    
=MESSAGEBOX(lcConditionsFailMessage,16, ;

        
THIS.Name+" kann nicht ausgeführt werden")



THIS.lError = .T.


ENDIF   


IF NOT THIS.lError



THIS.oWordRef.AddAddIn(THIS.cAddInFRXDOT,1)



THIS.oWordRef.FileNew(THIS.cDot)

        THIS.oWordRef.FileSaveAs(THIS.cTarget)


ENDIF


WAIT CLEAR

RETURN ! THIS.lError

PROCEDURE Setup2(tcAlias, tcScope, tnSession, tcTarget, tcDot)


IF TYPE("THIS.oWordRef") # "O" OR ISNULL(THIS.oWordRef)



RETURN "Kann nicht auf Word zugreifen!"


ENDIF


* Hier weggelassen: Setzen und Testen zusätzlicher


* Frx2Doc-spezifischer Bedingungen, die im 

    * Artikel beschrieben werden


* Gibt im Fehlerfall eine Meldung zurück.

RETURN ""

PROC Destroy


FRXClass::Destroy()


THIS.oWordRef = .NULL.

ENDPROC

PROC Run


LOCAL lnDocNumber


FRXClass::Run()


* Als erstes stellen sicher, daß Ihr Dokument aktiv ist. 


lnDocNumber = THIS.GetDocNumber(THIS.cTarget)


IF lnDocNumber > 0 



THIS.oWordRef.WindowList(lnDocNumber)



THIS.oWordRef.FileSave( )



THIS.oWordRef.FileClose(1)


ENDIF

ENDPROC

FUNC IsDocOpen(tcTargetName)


LOCAL lnCount, liIndex, llFound


lnCount = THIS.oWordRef.CountWindows()


IF lnCount > 0



FOR liIndex = 1 TO lnCount




IF UPPER(tcTargetName) == ;





UPPER(THIS.oWordRef.FileNameFromWindow(liIndex))





llFound = .T.





EXIT




ENDIF



ENDFOR


ENDIF      

RETURN llFound

FUNC GetDocNumber(tcTarget)


LOCAL lnCount, liIndex, lnFound


lnCount = THIS.oWordRef.CountWindows()


lnFound = 0


IF lnCount > 0



FOR liIndex = 1 TO lnCount




IF UPPER(tcTarget) == ;





UPPER(THIS.oWordRef.FileNameFromWindow(liIndex))





lnFound = liIndex





EXIT




ENDIF



ENDFOR


ENDIF      

RETURN lnFound

* Alle Methoden, die im Ereigniscode der FRX benötigt werden

* bleiben in der abstrakten Klasse Frx2Doc leer, 

* Sie können sie für verschiedene Berichte individuell füllen

ENDDEFINE

Listing C:
Auszug aus FRXCLASS.PRG
DEFINE CLASS FRXClass AS Custom


cFRXFileDir = SET("DIRECTORY")


cTarget = THIS.cFRXFileDir + "\FrxClass.PRN"


cTargetExt = "PRN"


cVFPReport = "FRXClass.FRX"


cScope = ""


nOldSelect = 0


nOldSession = 0


lError = .F.


Name = "FRXClass"


cRunMessage = "Erstelle Dokument..."


cInitMessage = "Initialisiere Dokument..."


PROC Init(tcAlias, tcScope, tnSession, tcTarget)



WAIT WINDOW NOWAIT THIS.cInitMessage



LOCAL lcConditionsFailMessage 



THIS.cTarget = THIS.GetFileName(THIS.cTarget,tcTarget,THIS.cTargetExt)



* Ausführen und Testen der notwendigen Setup-Bedingungen:

      
lcConditionsFailMessage = ;




THIS.Setup(tcAlias, tcScope, tnSession, THIS.cTarget)



IF NOT EMPTY(lcConditionsFailMessage)




=MESSAGEBOX(lcConditionsFailMessage,16, ;





THIS.Name+" kann nicht ausgeführt werden")




THIS.lError = .T.



ENDIF   



WAIT CLEAR


RETURN ! THIS.lError


PROC Setup(tcAlias, tcScope, tnSession, tcTarget)



IF NOT _WINDOWS




RETURN "Dieser Code ist nicht plattformübergreifend!"




* Zum Beispiel werden kein Pfade auf dem Mac behandelt...



ENDIF



* Hier ausgelassen: Zusätzliche Fehlerbehandlung,



* Setzen von Datensitzung, Alias und Bereich


RETURN ""


PROC Destroy



WAIT CLEAR



IF NOT THIS.lError




=MESSAGEBOX(THIS.Name +" erstellt Ihren Bericht als"+;





CHR(13)+THIS.cTarget+".",0, ;





THIS.Name+" Finished!")



ENDIF                     



*
Hier ausgelassen: zurücksetzen von Datensitzung und Alias 



RELEASE oFRX


ENDPROC


PROC Run



LOCAL lcScopeString



WAIT WINDOW NOWAIT THIS.cRunMessage



lcScopeString = THIS.cScope   



REPORT FORM (THIS.cVFPReport) ;




&lcScopeString ;




NOCONSOLE ;




NAME oFRXDE



RELEASE THIS                  


RETURN


FUNC GetFileName(tcDefaultFileName, ;



tcDesiredFileName, ;



tcExtension)



* Ausgelassen: Setzen eines gültigen Dateinamens und



* eines Pfades 



* Automation Server benötigen einen vollqualifizierten Pfad


RETURN lcReturn


PROC Error(nError, cMethod, nLine)



THIS.lError = .T.



* Hier nur rudimentär...



WAIT WINDOW cMethod+CHR(13)+STR(nError)+;




CHR(13)+MESSAGE( )+CHR(13)+;




MESSAGE(1)


ENDPROC


PROC UpdateMessage



* Hier ausgelassen: Rückmeldung an den



* Anwender, Beschreibung im Artikel


ENDPROC                         


* Gruppierungs-Bedingungen für jede Stufe der Gruppierung:


FUNC GroupLevel1


RETURN .T.


* Hier ausgelassen: zusätzliche Gruppierungsbedingungen


* Leere Methoden 


PROC TitleEntry()


ENDPROC 


PROC TitleExit()


ENDPROC 


*  Hier ausgelassen: Funktionen für zusätzliche Gruppierungen


FUNC DEInit(toDE)


RETURN .T.


*  Hier ausgelassen: Zusätzliche DE-Methoden   

ENDDEFINE

Listing D -
Auszug aus FRXCLASS.PRG
Die Listings C und D zeigen Ihnen Aufzüge der Definitionen von Frx2Doc und FrxClass. Die Klasse Frx2Doc erhält mehrere Parameter, die ersten vier werden von der Superklasse FrxClass vorgeschrieben, die fünfte ist WinWord- und Frx2Doc-spezifisch. Die Parameter enthalten folgende Angaben:

· den Alias zum SELECTen, der die treibende Kraft für die Bewegung des Datensatzzeigers durch das FRX ist,

· den Bereich, in dem das FRX sich in der SELECTierten Tabelle bewegen soll,

· die Datensitzung, in der der Alias geöffnet ist, auf den zugegriffen wird,

· das zu erstellende Dokument für die Ausgabe,

· die .DOT-Datei, die benutzt werden soll.

Das Programm, das eine von Frx2Doc abgeleitete Klasse instantiiert, sollte wie dieses aussehen:

PUBLIC oFRX

oFRX = CREATEOBJECT(<von Frx2Doc abgeleitete Klasse>)


tcAlias, tcScope, ;


tnSession, ;


tcTarget, tcDot, ;


<weitere Parameter>

IF TYPE(“oFRX“) = “O“ AND NOT ISNULL(oFRX)


oFRX.Run()

ENDIF
Beachten Sie, daß oFRX ein unbedingt erforderlicher Variablenname ist, da diese Referenz für alle Ereignisaufrufe in der FRX benötigt wird. Das Objekt oFRX räumt auf und gibt in seiner Methode Run() selbst frei.

Ihre Methode Run() könnte, bevor sie etwas anderes macht, versuchen, Word mit dem Dokument in der Seitenansicht zu öffnen. Sie könnte auch QUICKVIEW, den Dateibetrachter, der mit Win95 ausgeliefert wird, starten, um dem Anwender die Möglichkeit zu geben, sich das Dokument anzusehen. Wie ich bereits früher beschrieben habe, könnte Run() es dem Anwender auch ermöglichen, das Dokument in einem beliebigen Dateityp zu speichern. Sie können diese Vorschläge noch beliebig erweitern Sie finden genauere Erläuterungen zu diesem Vorgehen, dem Gebrauch und Entwurf der Klassen Frx2Doc und FrxClass in den vollständigen Source, die die Klassendefinitionen enthalten (Listing C und D: FRX2DOC.PRG).

Wenn Sie sich die generische Natur von FrxClass ansehen, werden Sie leicht verstehen, wie das entsprechende Berichtsformular andere Ausgabeformate steuern kann. Stellen Sie sich zum Beispiel vor, es wird eine Präsentation von PowerPoint mit zwei Tabellen in einer 1:n-Beziehung gesteuert, von denen die Mastertabelle Überschriften oder Titel enthält und die Detailtabelle einzelne Punkte oder die Darstellungen selber enthält. 

Sie starten die OLE-Sitzung mit dem bekannten CREATEOBJECT() und benutzen dann das Objektmodell von PowerPoint, fügen der Sammlung der Präsentationen ein Mitglied hinzu und dem neuen Präsentationsobjekt Folien hinzu. Der Code in der Gruppe könnte zum Beispiel so aussehen:

oSlide = oPresent.Slides.Add(<number>,<layout>)

oSlide.Objects.Title.Text = ALLTRIM(<Feldname>)

*
Erstellen einer Referenz auf den „Körper der Folie

oBody = oSlide.Objects.Objects.Placeholders(1)

Der Code für den Detailbereich geht weiter mit dem füllen der Objektreferenz oBody mit zusammengesetzten Zeichenketten, die die Punkte, die Sie auf der Folie darstellen wollen, benennen. Sie werden durch Zeilenumbrüche voneinander getrennt. Aufbauend auf den Layoutinformationen, die Sie ausgewählt haben, werden diese Punkte automatisch formatiert. Die Formatanweisungen können sie aber ändern und Grafiken, eigene Hintergründe usw. „während des laufenden Betriebs“, ganz wie es Ihre Daten erfordern hinzufügen.

Jenseits von OLE

Nachdem Sie nun einmal gesehen haben, wie OLE Automation arbeitet, ist der Code, den Sie für den jeweiligen Server schreiben, einfach und vorhersehbar. Je offener das Objektmodell des Server, desto einfacher und vorhersehbarer ist der Code. Daher werde ich auch keinen extra Artikel schreiben, in dem ich Ihnen zeige, wie Sie den Code einer Klasse für PowerPoint schreiben, die Sie von FrxClass ableiten. Außerdem steht Ihnen mit Frx2Doc eine gut entwickelte von FrxClass abgeleitete Klasse zur Verfügung. Diese Beispiele sollten Ihnen auch helfen, mit jedem anderen OLE Automation Server zu arbeiten.

Allerdings definiert OLE Automation noch nicht die Grenzen von Nicht-VFP-Ausgabemöglichkeiten oder die Möglichkeiten von FrxClass. In der nächsten Folge dieses Serie, werden wir mit FrxClass HTML-formatierten Text generieren, mit dem Sie WWW-Dokumente erstellen können. Sie werden sehen, daß die gleiche „Technologie des leeren Berichts“, mit der wir hier ein WinWord-Dokument erstellt haben, auch in ein vollständig anderes Ziel ausgeben kann.

